

PROPOSED ALGORITHM FOR DETERMINING THE DELTA INTERCEPT OF A THERMOCOUPLE PSYCHROMETER CURVE

U.S. GEOLOGICAL SURVEY

Open-File Report 92-490

HYDROLOGY DOCUMENT NUMBER 715

Prepared in cooperation with the
U.S. DEPARTMENT OF ENERGY
(Interagency Agreement DE-AI08-92NV10874)



**PROPOSED ALGORITHM FOR DETERMINING THE DELTA
INTERCEPT OF A THERMOCOUPLE PSYCHROMETER
CURVE**

by Mark A. Kurzmack

U.S. GEOLOGICAL SURVEY

Open-File Report 92-490

Prepared in cooperation with the
U.S. DEPARTMENT OF ENERGY
(Interagency Agreement DE-AI08-92NV10874)

Denver, Colorado
1993



U.S. DEPARTMENT OF THE INTERIOR

BRUCE BABBITT, Secretary

U.S. GEOLOGICAL SURVEY

Dallas L. Peck, Director

For additional information
write to:

Chief, Hydrologic Investigations Program
Yucca Mountain Project Branch
U.S. Geological Survey
Box 25046, Mail Stop 421
Denver Federal Center
Denver, CO 80225-0046

Copies of this report can
be purchased from:

U.S. Geological Survey
Books and Open-File Reports Section
Box 25425, Mail Stop 517
Denver Federal Center
Denver, CO 80225-0425

CONTENTS

	Page
Abstract-----	1
Introduction-----	1
Purpose and scope-----	1
Thermocouple psychrometer methodology-----	1
Algorithm overview-----	2
Implementation of the algorithm in a Pascal procedure-----	4
Pascal source code for calculate_intercept procedure-----	5
Conclusion-----	8
References cited-----	8

FIGURES

Figure 1. Components of a thermocouple psychrometer voltage output curve-----	1
2. Stepping the trial regression line along the psychrometer curve-----	3
3. Plot of the trial intercepts calculated by the procedure-----	4
4. Final calculation of the delta intercept-----	4

Proposed Algorithm for Determining the Delta Intercept of a Thermocouple Psychrometer Curve

By Mark A. Kurzmack

ABSTRACT

An algorithm is proposed for simply and efficiently reducing a six wire thermocouple psychrometer voltage output curve to a single value - the delta intercept. The algorithm identifies a plateau region in the psychrometer curve and extrapolates a linear regression back to the initial start of relaxation. When properly conditioned for the measurements being made, the algorithm results in reasonable results even with incomplete or noisy psychrometer curves over a 1 to 60 bar range.

INTRODUCTION

Purpose and Scope

The USGS Hydrologic Investigations Program is currently developing instrumentation to study the unsaturated zone at Yucca Mountain in Nevada. Surface-based boreholes up to 2500 feet in depth will be drilled, and then instrumented in order to define the water potential field within the unsaturated zone. Thermocouple psychrometers will be used to monitor the in-situ water potential. This paper describes an algorithm for simply and efficiently reducing a six-wire thermocouple psychrometer voltage output curve to a single value - the delta intercept.

Thermocouple Psychrometer Methodology

Thermocouple psychrometers are very sensitive instruments widely used for measuring water potential (Brown, 1970; Brown and Van Haveren, 1972; Wiebe and others, 1971). A representative six-wire thermocouple psychrometer [designed by the United States Geological

Survey and built by J.R.D. Merrill Specialty Equipment¹, Logan, Utah] voltage output curve is shown in figure 1. The curve was obtained using a Keithley Model 181 nanovoltmeter to measure voltage, and a Keithley Model 220 current generator for excitation. The normal output consists of an initial increase in voltage, a plateau, and a final voltage increase back to the voltmeter zero level.




Figure 1.--Components of a thermocouple psychrometer voltage output curve.

The procedure for obtaining a thermocouple psychrometer reading consists of three components:

1. Determination of a voltmeter zero value;
2. Excitation of the psychrometer with a cooling current that causes condensation

¹Use of trade or firm names in this report is for identification purposes only and does not constitute endorsement by the U.S. Geological Survey.

of water vapor onto the thermocouple bead; and

3. Relaxation of the psychrometer from the excited state back to the voltmeter zero level as the bead warms up again and the water droplets evaporate.

When the thermocouple psychrometer is in thermal equilibrium with its environment, all the voltages generated by the thermocouple junctions in the psychrometer should cancel, and there should be no net voltage measured from the psychrometer. In practice, there may be a small residual voltage from the thermocouples, and additional contributions from any zero offsets in the nanovoltmeter or measuring electronics. This small voltage is referred to as the "voltmeter zero"

After the voltmeter zero is determined, a current is applied to the thermocouple psychrometer to cool the thermocouple (Peltier effect) and displace it from thermal equilibrium. The cooling causes water droplets to condense on the surface of the thermocouple bead. When the current is removed (at 4 seconds in figure 1), the thermocouple psychrometer warms up again. The rate of the warming is modified by the evaporation of the water droplets on the thermocouple bead, producing a slowing of the warming process (the plateau region) until all the water has evaporated. Once the water has evaporated, warming of just the bead continues.

As part of the Surface-Based Borehole Investigations of Yucca Mountain, Nevada, (which is being evaluated for use as a potential underground repository for high-level radioactive waste), it will be necessary to have more than 600 calibrated psychrometers available for measuring water potential. The quantity of sensors necessitates automating the calibration process as much as possible, and makes it impractical to deal with the complete psychrometer curves. The data from each curve must be reduced to a single value for use in calibrating the psychrometers. Therefore, a method was selected to extrapolate the voltage plateau back to the start of the decay process in order to

provide an intercept value that corresponds to the initial point of evaporation (Savage and Wiebe, 1987). The difference between the intercept and the voltmeter zero will be called the delta intercept.

By measuring the delta intercept values generated at several temperatures with a range of sodium chloride solutions that correspond to various water potentials in a laboratory, it is possible to provide a calibration for the delta intercept values measured in the field. The accuracy of the measured delta intercept is dependent on both the determination of the voltmeter zero and on the extrapolation of the regression line.

ALGORITHM OVERVIEW

The difficulty in analyzing the thermocouple psychrometer voltage output curve to determine the delta intercept lies in identification of the plateau region in a systematic and reproducible fashion. Although it may appear trivial to select a plateau visually, automating the process is not straightforward when the entire curve is not available, or when the part of the curve that is available has a considerable amount of noise.

In addition, the actual shape of the psychrometer curve changes when going from dry to wet conditions. A method that may be ideal for the well-defined curve shown in figure 1 may fail at very wet or very dry conditions. Under very wet conditions, the plateau is so extended that the final increase in voltage may not occur for 10 minutes or longer. Wetter conditions also result in a weaker signal and increased signal-to-noise problems. Very dry conditions, on the other hand, may result in such a short and steep plateau that it becomes difficult to identify.

Rather than attempting to identify the entire plateau region between the initial voltage increase and the final voltage increase, it was decided to eliminate the short initial increase and then extrapolate a line drawn through the points following the initial increase back to the

start of the decay period. The use of only the initial portion of the plateau minimizes problems associated with drift or curvature in the plateau that might occur due to secondary effects such as uneven cooling or evaporation, or instrument drift when the measurement is extended for a long period of time. Since the initial voltage increase usually is fairly brief, enough of the curve is available to perform the calculations without having to monitor the psychrometer for a prolonged period, even under very wet conditions.

The algorithm attempts to select the start of the plateau. Plateaus may show considerable curvature that will result in different intercepts at different parts of the plateau. However, the start of the plateau is closest to the conditions that exist when excitation ends, and requires the least extrapolation of the regression line.

The proposed algorithm steps along the curve from the beginning and calculates linear regressions for sequential subsets of the thermocouple psychrometer curve. The process is simulated in figure 2. The numbered lines represent sequential trial regressions along the curve. Line 1 represents the linear regression through the first 30 points of the curve. Due to the inclusion of the points in the initial voltage increase, the line has a steep slope and the trial intercept is very negative. As the subset is moved along the curve, the first point is dropped off, and a point is added to the end to give line 2. Since less of the initial voltage increase is included in the regression, the line is not as steep as line 1, and the trial intercept is not as negative. The intercept will continue moving upward until the regression has reached the plateau (line 4). At that point, the trial intercepts will oscillate around the true intercept due to the noise in the points in the plateau.

If one continued to step along the thermocouple psychrometer curve in this fashion, the trial intercepts eventually would become more negative again as points from the final increase begin to be included. A plot of the trial intercepts versus the starting time of the sequential subsets used for the linear regressions is shown




Figure 2.--Stepping the trial regression line along the psychrometer curve. The numbered lines represent sequential trial regressions along the curve. Line 4 best represents the plateau region.

in Figure 3. There is a quick increase in the delta intercept and then a flattening out as the plateau is reached. The point used to calculate the actual delta intercept is indicated in figure 3. The original curve is shown in Figure 4. The points located between sample start and sample finish were used to calculate the delta intercept by extrapolating the intercept regression line back to the time at which excitation ended (4 seconds). To compensate for noise in the curve and for the fact that the plateau may not be perfectly flat, the curve is first smoothed by replacing the measured values with median values. The number of points used in each regression also is adjusted to compensate for the shorter plateaus observed with dry samples. The number of points has been adjusted empirically by comparing the results of the algorithm with lines drawn visually for a 1 to 60 bar equivalent range of salt solutions for the instruments and psychrometer type being used.




Figure 3.--Plot of the trial intercepts calculated by the procedure.

In order to determine the actual delta intercept, the algorithm selects the highest (least negative) intercept. The algorithm continues to step along the curve until enough regressions with equal or lower (more negative) intercepts than the trial delta intercept are found. Ten points with intercepts lower than the trial point are required to satisfy the stopping criteria. This requirement insures that the process has stepped up far enough on the curve to reach the plateau region (see fig. 3).

In practice, the algorithm works best when there is a clearly defined plateau region. When the plateau is ill defined due to noise, shows

curvature, or does not contain an adequate number of data points, the correct region of the curve may not be selected. If the curve has a continuously decreasing first derivative without a definite plateau, the calculated intercepts will continue to slowly increase in value, and the algorithm will continue to step up the curve indefinitely. For very dry conditions, the plateau may be so short that the algorithm may




Figure 4.--Final calculation of the delta intercept. The area bounded by sample start and sample finish is the section of the curve used for the calculation of the regression line and the delta intercept.

miss it entirely. If these conditions exist and the algorithm runs out of data without meeting the stopping criteria, the value of the delta intercept calculated at the first trial point is used as a default and the failure is indicated. To properly condition the algorithm, a wide range of conditions should be tested to empirically determine the proper sample size for the conditions and type of psychrometer to be used.

Implementation of the Algorithm in a Pascal Procedure

The source code for a procedure for determining the delta intercept of a thermocouple psychrometer curve follows this description. The procedure is written in Turbo Pascal 5.5 (Borland International, Scotts Valley, California) and uses some library routines from Turbo Professional 5.0 (TurboPower Software, Scotts Valley, California).

Pascal Source Code for Calculate Intercept Procedure

PROCEDURE calculate_intercept(numpts : word);
{numpts is the number of data points in the curve}

{Global variables referenced:

delta_intercept	- the calculated intercept of the line drawn through the plateau
delta_slope	- the slope of the line drawn through the plateau
intercept_error_flag	- normally false, but set true if algorithm fails
time[i]	- the time at which the ith voltage was measured
voltmeterzero	- the value of the voltmeter zero voltage
volts[i]	- the ith raw voltage in the relaxation curve}

CONST

tolerance = 0.0005E-6; {scatter allowed in intercept calculations}

VAR

a	: ARRAY[1..250] Of REAL; {temporary array}
v	: ARRAY[1..250] Of REAL; {array of corrected voltages}
z	: ARRAY[1..250] Of REAL; {smoothed psychrometer curve}
inter	: ARRAY[1..250] Of REAL; {array of intercepts}
slope	: ARRAY[1..250] Of REAL; {array of slopes}
temp	: REAL;
s0, s3, s4, s7, s9	: REAL; {intermediate values in regression calc}
i, j, k	: INTEGER;
maxnum	: INTEGER; {number of data points considered by the algorithm (maximum is 250)}
criterion	: INTEGER; {number of regression lines that must have intercepts less negative than the trial intercept in order to stop the process}
position	: INTEGER; {current position on curve}
possmin	: INTEGER; {# pt which is a potential minimum}
done_flag	: BOOLEAN; {true when stopping criteria are met}
sampsize	: WORD; {number of points used to calculate regression line}

BEGIN

```
WRITELN('Now determining intercept ...');  
intercept_error_flag := FALSE; {set true if algorithm fails}  
done_flag := FALSE;  
criterion := 10;  
maxnum := minword(numpts, 250); {finds smaller of two arguments}  
FOR i := 1 To maxnum DO v[i] := volts[i] - voltmeterzero;
```

```
{smooth using a running median of 4 followed by 2}
```

```
FOR j := 3 To maxnum - 2 DO
```

```
  BEGIN
```

```
    FOR i := j - 2 To j + 2 DO a[i] := v[i];  
      {exchangestructs will exchange the two values}  
    FOR i := j - 1 To j + 1 DO  
      IF a[i] < a[j - 2] THEN exchangestructs(a[i], a[j - 2], 6);  
      IF a[j - 1] > a[j + 1] THEN exchangestructs(a[j - 1], a[j + 1], 6);  
      IF a[j] > a[j + 1] THEN exchangestructs(a[j], a[j + 1], 6);  
      z[j] := (a[j - 1] + a[j]) / 4.0; {average of first median}  
      FOR i := j - 2 To j + 2 DO a[i] := v[i];  
      FOR i := j To j + 2 DO
```

```

    IF a[i] < a[j - 1] THEN exchangestructs(a[i], a[j - 1], 6);
    IF a[j] > a[j + 2] THEN exchangestructs(a[j], a[j + 2], 6);
    IF a[j + 1] > a[j + 2] THEN exchangestructs(a[j + 1], a[j + 2], 6);
    z[j] := z[j] + (a[j] + a[j + 1]) / 4.0; {add average of second median}
END;

```

```
{select sample size}
```

```

temp := 0;
FOR i := 4 to 11 DO temp := temp + z[i];
temp := ABS(temp * 125000.0); {convert to microvolts and average}
if temp < 3.0 then possmin := 20 ELSE possmin := 3;
sampsiz := ROUND(4 + EXP(-temp / 8.0) * 145 - 0.4 * temp);
if sampsiz < 4 then sampsiz := 4;

```

```
{now start stepping along curve}
```

```
position := 3; delta_intercept := 0; delta_slope := 0;
```

```
{check to see if the first trial point is best intercept}
```

```
REPEAT
```

```
IF position <= maxnum - sampsiz - criterion THEN
```

```
  BEGIN
```

```
    s3 := 0; s4 := 0; s7 := 0; s9 := 0;
```

```
    FOR j := position To position + sampsiz - 1 DO
```

```
      BEGIN
```

```
        {linear regression calculations}
```

```
        s3 := s3 + time[j]; s4 := s4 + time[j] * time[j];
```

```
        s7 := s7 + z[j]; s9 := s9 + time[j] * z[j];
```

```
      END;
```

```
    s0 := sampsiz * s4 - s3 * s3;
```

```
    IF s0 <> 0 THEN
```

```
      BEGIN
```

```
        inter[position] := (s4 * s7 - s3 * s9) / s0;
```

```
        slope[position] := (sampsiz * s9 - s3 * s7) / s0;
```

```
      END;
```

```
    IF position < possmin + criterion THEN
```

```
      INC(position)
```

```
    ELSE
```

```
      BEGIN
```

```
        k := possmin + 1;
```

```
        {continue stepping along curve until criterion points are  
less than inter[possmin]+tolerance or test fails}
```

```
        WHILE (inter[k] <= inter[possmin]+tolerance)
```

```
          AND (done_flag = FALSE) DO
```

```
            IF k < possmin + criterion
```

```
              THEN INC(k)
```

```
            ELSE done_flag := True;
```

```
          {if test failed move possmin and increment position}
```

```
          IF done_flag = FALSE THEN
```

```
            BEGIN
```

```
              possmin := k;
```

```
              INC(position);
```

```
            END;
```

```
          END;
```

```
        END
```

```

ELSE
  BEGIN {Algorithm failed - just use first possmin value}
    WRITELN('Unable to determine intercept normally');
    WRITELN('Intercept calculated at initial trial point');
    IF temp < 3.0 then possmin := 20 ELSE possmin := 3;
    intercept_error_flag := True;
    done_flag := True;
  END;
UNTIL done_flag = True;
delta_intercept := inter[possmin];
delta_slope := slope[possmin];
WRITELN('Intercept: ', delta_intercept * 1e6:8:3, 'microvolts');
sampstart := possmin;
END;

```

Synopsis of Procedure:

1. Initialize variables
 - critterion = 10 This may be adjusted if necessary
 - intercept_error_flag = false
Set true if the algorithm fails to terminate normally
 - maxnum = the smaller of 250 or the actual number of points in the curve corrected voltage $v[i] = \text{volts}[i] - \text{voltmeterzero}$
2. Smooth using a running median of four followed by two. The smoothed curve is placed in $z[i]$ where i ranges from 3 to $(\text{maxnum} - 2)$. (Velleman and Hoaglin, 1981)
3. Select the sample size
 - The sample size is adjusted to compensate for drier curves that have a shorter plateau region. This is done by averaging points 4 to 11 (which are likely to be close to the plateau), converting the average to microvolts, and calculating the sample size based on a modified decaying exponential equation selected empirically to provide a reasonable sample size with a selection of different psychrometers over a range of salt solu-

tions. The minimum number of points is 4. The sample sizes calculated in this way have been found to be effective for 1 to 60 bar equivalent salt solutions. The calculation of the sample size should be adjusted to account for differences in response between different types of psychrometers and measurement instruments.

4. Now start stepping along the curve

Start at position=3 with the number of the first trial point (possmin) equal to 3 for dry solutions or 20 for wet solutions. Take a linear regression of the points from position to $(\text{position} + \text{sampsiz} - 1)$ and calculate the intercept and slope of the regression line. If the position is less than the value needed to satisfy the stopping criteria at the first possible point $(\text{possmin} + \text{critterion})$, just increment the position and loop back to take another regression at the new point. At that point, a long enough baseline has been calculated to check the regression values against the stopping criteria.

The stopping criteria require that a number (critterion) of intercepts be less than (have a more negative voltage than) the intercept calculated at the trial point (possmin). To allow for some scatter in the calculated intercepts, the value of

tolerance can be set to some small positive value and added to the intercept calculated at the trial point. The use of a tolerance value allows a point that is just slightly greater than the possmin intercept due to scatter to still be used in satisfying the stopping criteria. If the conditions are satisfied, possmin becomes the solution point. If the conditions are not satisfied, both position and possmin are incremented, and we continue to step along the curve and test for the stopping criteria.

If we step along the curve and are unable to satisfy the stopping criteria before the last possible calculation point (maximum-samplesize-criterion) is reached, the algorithm has failed to determine the delta intercept normally. This failure may either be reported as a bad value, or the initial intercept calculated at the initial value of possmin may be used as an approximate value.

CONCLUSION

The proposed algorithm provides a computationally simple and efficient method for reducing a psychrometer curve to a single voltage value. The algorithm works over a wide range of water vapor saturation conditions (1 to 60 bars) and permits the data reduction to proceed with incomplete or noisy curves when properly conditioned.

REFERENCES CITED

Brown, R.W., 1970, Measurement of water potential with thermocouple psychrometers - Construction and applications: United States Department of Agriculture Forest Service Research Paper INT-80, 27 p. (NNA.900208.0001)

Brown, R.W., and Van Haveren, B.P., eds., 1972, Psychrometry in water relations research: Proceedings of the Symposium on Thermocouple Psychrometers: Utah Agricultural Experiment Station, Utah State University, 342 p. (NNA.921211.0164)

Savage, M.J., and Wiebe, H.H., 1987, Voltage endpoint determination for thermocouple psychrometers and the effect of cooling time: Agricultural and Forest Meteorology, v. 39, p. 309-317. (NNA.921130.0003)

Velleman, P.F., and Hoaglin, D.C., 1981, Applications, basics, and computing of exploratory data analysis: Boston, Duxbury Press, p. 159-190. (NNA.921211.0172)

Wiebe, H.H., Campbell, G.S., Gardner, W.H., Rawlins, S.L., Cary, J.W., and Brown, R.W., 1971, Measurement of plant and soil water status: Utah Agricultural Experiment Station, Utah State University, Bulletin 484, 71 p. (NNA.921222.0011)

NOTE: Parenthesized numbers following each cited reference are for U.S. Office of Civilian Radioactive Waste Management Records management purposes only and should not be used when ordering the publication.

The following number is for U.S. Department of Energy Office of Civilian Radioactive Waste Management Records management purposes only, and should not be used when ordering this publication: Accession number-- NNA.930104.0001

